A Shorter Compiler Correctness Proof for Language IMP

Pasquale Noce
Software Engineer at HID Global, Italy
pasquale dot noce dot lavoro at gmail dot com
pasquale dot noce at hidglobal dot com

March 17, 2025

Abstract

This paper presents a compiler correctness proof for the didactic imperative programming language IMP, introduced in Nipkow and Klein's book on formal programming language semantics (version of March 2021), whose size is just two thirds of the book's proof in the number of formal text lines. As such, it promises to constitute a further enhanced reference for the formal verification of compilers meant for larger, real-world programming languages.

The presented proof does not depend on language determinism, so that the proposed approach can be applied to non-deterministic languages as well. As a confirmation, this paper extends IMP with an additional non-deterministic choice command, and proves compiler correctness, viz. the simulation of compiled code execution by source code, for such extended language.

Contents

1 Compiler formalization

theory *Compiler*
imports
HOL−IMP.BExp

HOL−IMP.Star
begin

1.1 Introduction

This paper presents a compiler correctness proof for the didactic imperative programming language IMP, introduced in [5], shorter than the proof described in [5] and included in the Isabelle2021 distribution [1]. Actually, the size of the presented proof is just two thirds of the book's proof in the number of formal text lines, and as such it promises to constitute a further enhanced reference for the formal verification of compilers meant for larger, real-world programming languages.

Given compiler *completeness*, viz. the simulation of source code execution by compiled code, "in a deterministic language like IMP", compiler correctness "reduces to preserving termination: if the machine program terminates, so must the source program", even though proving this "is not much easier" ([5], section 8.4). However, the presented proof does not depend on language determinism, so that the proposed approach is applicable to non-deterministic languages as well.

As a confirmation, this paper extends IMP with an additional command c_1 *OR* c_2, standing for the non-deterministic choice between commands c_1 and c_2, and proves compiler *correctness*, viz. the simulation of compiled code execution by source code, for such extended language. Of course, the aforesaid comparison between proof sizes does not consider the lines in the proof of lemma *ccomp-correct* (which proves compiler correctness for commands) pertaining to non-deterministic choice, since this command is not included in the original language IMP. Anyway, non-deterministic choice turns out to extend that proof just by a modest number of lines.

For further information about the formal definitions and proofs contained in this paper, see Isabelle documentation, particularly [6], [4], [2], and [3].

1.2 Definitions

Here below are the definitions of IMP commands, extended with non-deterministic choice, as well as of their big-step semantics.

As in the original theory file [1], program counter's values are modeled using type *int* rather than *nat*. As a result, the same declarations and definitions used in [1] to deal with this modeling choice are adopted here as well.

declare [[*coercion-enabled*]]
declare [[*coercion int* :: *nat* ⇒ *int*]]
declare [[*syntax-ambiguity-warning* = *false*]]

datatype *com* =
SKIP |
Assign vname aexp (‹- ::= ->› [*1000*, *61*] *61*) |
Seq com com (‹-;;/ ->› [*60*, *61*] *60*) |
If bexp com com (‹(*IF* -/ *THEN* -/ *ELSE* -)› [*0*, *0*, *61*] *61*) |
Or com com (‹(- *OR* -)› [*60*, *61*] *61*) |
While bexp com (‹(*WHILE* -/ *DO* -)› [*0*, *61*] *61*)

inductive *big-step* :: *com* × *state* ⇒ *state* ⇒ *bool* (**infix** ‹⇒› *55*) **where**
Skip: (*SKIP*, *s*) ⇒ *s* |
Assign: (*x* ::= *a*, *s*) ⇒ *s*(*x* := *aval a s*) |
Seq: ⟦(*c*$_1$, *s*$_1$) ⇒ *s*$_2$; (*c*$_2$, *s*$_2$) ⇒ *s*$_3$⟧ ⟹ (*c*$_1$;; *c*$_2$, *s*$_1$) ⇒ *s*$_3$ |
IfTrue: ⟦*bval b s*; (*c*$_1$, *s*) ⇒ *t*⟧ ⟹ (*IF b THEN c*$_1$ *ELSE c*$_2$, *s*) ⇒ *t* |
IfFalse: ⟦¬ *bval b s*; (*c*$_2$, *s*) ⇒ *t*⟧ ⟹ (*IF b THEN c*$_1$ *ELSE c*$_2$, *s*) ⇒ *t* |
Or1: (*c*$_1$, *s*) ⇒ *t* ⟹ (*c*$_1$ *OR c*$_2$, *s*) ⇒ *t* |
Or2: (*c*$_2$, *s*) ⇒ *t* ⟹ (*c*$_1$ *OR c*$_2$, *s*) ⇒ *t* |
WhileFalse: ¬ *bval b s* ⟹ (*WHILE b DO c*, *s*) ⇒ *s* |
WhileTrue: ⟦*bval b s*$_1$; (*c*, *s*$_1$) ⇒ *s*$_2$; (*WHILE b DO c*, *s*$_2$) ⇒ *s*$_3$⟧ ⟹
(*WHILE b DO c*, *s*$_1$) ⇒ *s*$_3$

declare *big-step.intros* [*intro*]

abbreviation (**output**)
isize xs ≡ *int* (*length xs*)

notation *isize* (‹*size*›)

primrec (*nonexhaustive*) *inth* :: *'a list* ⇒ *int* ⇒ *'a* (**infixl** ‹!!› *100*) **where**
(*x* # *xs*) !! *i* = (*if i* = *0 then x else xs* !! (*i* − *1*))

lemma *inth-append* [*simp*]:
0 ≤ *i* ⟹
(*xs* @ *ys*) !! *i* = (*if i* < *size xs then xs* !! *i else ys* !! (*i* − *size xs*))
⟨*proof*⟩

Next, the instruction set and its semantics are defined. Particularly, to allow for the compilation of non-deterministic choice commands, the instruction set is extended with an additional instruction *JMPND* performing a non-deterministic jump – viz. as a result of its execution, the program counter unconditionally either jumps by the specified offset, or just moves to the next instruction.

As instruction execution can be non-deterministic, an inductively defined predicate *iexec*, rather than a simple non-recursive function as the one used in [1], must be introduced to define instruction semantics.

datatype *instr* =
LOADI int | *LOAD vname* | *ADD* | *STORE vname* |

JMP int | *JMPLESS int* | *JMPGE int* | *JMPND int*

type-synonym *stack* = *val list*
type-synonym *config* = *int* × *state* × *stack*

abbreviation *hd2 xs* ≡ *hd* (*tl xs*)
abbreviation *tl2 xs* ≡ *tl* (*tl xs*)

inductive *iexec* :: *instr* × *config* ⇒ *config* ⇒ *bool* (**infix** ‹↦› 55) **where**
LoadI: (*LOADI i*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, *i* # *stk*) |
Load: (*LOAD x*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, *s x* # *stk*) |
Add: (*ADD*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, (*hd2 stk* + *hd stk*) # *tl2 stk*) |
Store: (*STORE x*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*(*x* := *hd stk*), *tl stk*) |
Jmp: (*JMP i*, *pc*, *s*, *stk*) ↦ (*pc* + *i* + *1*, *s*, *stk*) |
JmpLessY: *hd2 stk* < *hd stk* ⟹
(*JMPLESS i*, *pc*, *s*, *stk*) ↦ (*pc* + *i* + *1*, *s*, *tl2 stk*) |
JmpLessN: *hd stk* ≤ *hd2 stk* ⟹
(*JMPLESS i*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, *tl2 stk*) |
JmpGeY: *hd stk* ≤ *hd2 stk* ⟹
(*JMPGE i*, *pc*, *s*, *stk*) ↦ (*pc* + *i* + *1*, *s*, *tl2 stk*) |
JmpGeN: *hd2 stk* < *hd stk* ⟹
(*JMPGE i*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, *tl2 stk*) |
JmpNdY: (*JMPND i*, *pc*, *s*, *stk*) ↦ (*pc* + *i* + *1*, *s*, *stk*) |
JmpNdN: (*JMPND i*, *pc*, *s*, *stk*) ↦ (*pc* + *1*, *s*, *stk*)

declare *iexec.intros* [*intro*]

inductive-cases *LoadIE* [*elim*!]: (*LOADI i*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *LoadE* [*elim*!]: (*LOAD x*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *AddE* [*elim*!]: (*ADD*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *StoreE* [*elim*!]: (*STORE x*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *JmpE* [*elim*!]: (*JMP i*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *JmpLessE* [*elim*!]: (*JMPLESS i*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *JmpGeE* [*elim*!]: (*JMPGE i*, *pc*, *s*, *stk*) ↦ *cf*
inductive-cases *JmpNdE* [*elim*!]: (*JMPND i*, *pc*, *s*, *stk*) ↦ *cf*

definition *exec1* :: *instr list* ⇒ *config* ⇒ *config* ⇒ *bool*
(‹(-/ ⊢/ -/ →/ -)› 55) **where**
P ⊢ *cf* → *cf'* ≡ (*P* !! *fst cf*, *cf*) ↦ *cf'* ∧ *0* ≤ *fst cf* ∧ *fst cf* < *size P*

abbreviation *exec* :: *instr list* ⇒ *config* ⇒ *config* ⇒ *bool*
(‹(-/ ⊢/ -/ →*/ -)› 55) **where**
exec P ≡ *star* (*exec1 P*)

Next, compilation is formalized for arithmetic and boolean expressions (functions *acomp* and *bcomp*), as well as for commands (function *ccomp*). Particularly, as opposed to what happens in [1], here *bcomp* takes a single input, viz. a 3-tuple comprised of a boolean expression, a flag, and a jump off-

set. In this way, all three functions accept a single input, which enables to streamline the compiler correctness proof developed in what follows.

```
primrec acomp :: aexp ⇒ instr list where
acomp (N i) = [LOADI i] |
acomp (V x) = [LOAD x] |
acomp (Plus a1 a2) = acomp a1 @ acomp a2 @ [ADD]

fun bcomp :: bexp × bool × int ⇒ instr list where
bcomp (Bc v, f, i) = (if v = f then [JMP i] else []) |
bcomp (Not b, f, i) = bcomp (b, ¬ f, i) |
bcomp (And b1 b2, f, i) =
  (let cb2 = bcomp (b2, f, i);
     cb1 = bcomp (b1, False, size cb2 + (if f then 0 else i))
   in cb1 @ cb2) |
bcomp (Less a1 a2, f, i) =
  acomp a1 @ acomp a2 @ (if f then [JMPLESS i] else [JMPGE i])

primrec ccomp :: com ⇒ instr list where
ccomp SKIP = [] |
ccomp (x ::= a) = acomp a @ [STORE x] |
ccomp (c1;; c2) = ccomp c1 @ ccomp c2 |
ccomp (IF b THEN c1 ELSE c2) =
  (let cc1 = ccomp c1; cc2 = ccomp c2; cb = bcomp (b, False, size cc1 + 1)
   in cb @ cc1 @ JMP (size cc2) # cc2) |
ccomp (c1 OR c2) =
  (let cc1 = ccomp c1; cc2 = ccomp c2
   in JMPND (size cc1 + 1) # cc1 @ JMP (size cc2) # cc2) |
ccomp (WHILE b DO c) =
  (let cc = ccomp c; cb = bcomp (b, False, size cc + 1)
   in cb @ cc @ [JMP (− (size cb + size cc + 1))])
```

Finally, two lemmas are proven automatically (both seem not to be included in the standard library, though being quite basic) and registered for use by automatic proof tactics. In more detail:

- The former lemma is an elimination rule similar to *impCE*, with the difference that it retains the antecedent of the implication in the premise where the consequent is assumed to hold. This rule permits to have both assumptions $\neg$ *bval b s* and *bval b s* in the respective cases resulting from the execution of boolean expression b in state s.
- The latter one is an introduction rule similar to *Suc-lessI*, with the difference that its second assumption is more convenient for proving statements of the form $Suc\ m < n$ arising from the compiler correctness proof developed in what follows.

lemma *impCE2* [*elim*!]:
$[\![P \longrightarrow Q;\ \neg\ P \Longrightarrow R;\ P \Longrightarrow Q \Longrightarrow R]\!] \Longrightarrow R$
⟨*proof*⟩

lemma *Suc-lessI2* [*intro*!]:
$[\![m < n;\ m \neq n - 1]\!] \Longrightarrow Suc\ m < n$
⟨*proof*⟩

end

2 Compiler correctness

theory *Compiler2*
imports *Compiler*
begin

2.1 Preliminary definitions and lemmas

Now everything is ready for the compiler correctness proof. First, two predicates are introduced, *execl* and *execl-all*, both taking as inputs a program, i.e. a list of instructions, P and a list of program configurations *cfs*, and respectively denoted using notations $P \models cfs$ and $P \models cfs\Box$. In more detail:

- $P \models cfs$ means that program P *may* transform each configuration within *cfs* into the subsequent one, if any (word *may* reflects the fact that programs can be non-deterministic in this case study).
  Thus, *execl* formalizes the notion of a *small-step* program execution.
- $P \models cfs\Box$ reinforces $P \models cfs$ by additionally requiring that *cfs* be nonempty, the initial program counter be zero (viz. execution starts from the first instruction in P), and the final program counter falls outside P (viz. execution terminates).
  Thus, *execl-all* formalizes the notion of a *complete small-step* program execution, so that assumptions $acomp\ a \models cfs\Box$, $bcomp\ x \models cfs\Box$, $ccomp\ c \models cfs\Box$ will be used in the compiler correctness proofs for arithmetic/boolean expressions and commands.

Moreover, predicates *apred*, *bpred*, and *cpred* are defined to capture the link between the initial and the final configuration upon the execution of an arithmetic expression, a boolean expression, and a whole program, respectively, and abbreviation *off* is introduced as a commodity to shorten the subsequent formal text.

fun *execl* :: *instr list* ⇒ *config list* ⇒ *bool* (**infix** ‹⊨› *55*) **where**

$P \models cf \# cf' \# cfs = (P \vdash cf \to cf' \wedge P \models cf' \# cfs) \mid$
$P \models - = True$

definition *execl-all* :: *instr list* ⇒ *config list* ⇒ *bool* (‹(-/ ⊨/ -□)› 55) **where**
$P \models cfs\square \equiv P \models cfs \wedge cfs \neq [] \wedge$
$fst\ (cfs\ !\ 0) = 0 \wedge fst\ (cfs\ !\ (length\ cfs - 1)) \notin \{0..<size\ P\}$

definition *apred* :: *aexp* ⇒ *config* ⇒ *config* ⇒ *bool* **where**
$apred \equiv \lambda a\ (pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc' = pc + size\ (acomp\ a) \wedge s' = s \wedge stk' = aval\ a\ s \#\ stk$

definition *bpred* :: *bexp* × *bool* × *int* ⇒ *config* ⇒ *config* ⇒ *bool* **where**
$bpred \equiv \lambda(b,\ f,\ i)\ (pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc' = pc + size\ (bcomp\ (b,\ f,\ i)) + (if\ bval\ b\ s = f\ then\ i\ else\ 0) \wedge$
$s' = s \wedge stk' = stk$

definition *cpred* :: *com* ⇒ *config* ⇒ *config* ⇒ *bool* **where**
$cpred \equiv \lambda c\ (pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc' = pc + size\ (ccomp\ c) \wedge (c,\ s) \Rightarrow s' \wedge stk' = stk$

abbreviation *off* :: *instr list* ⇒ *config* ⇒ *config* **where**
$off\ P\ cf \equiv (fst\ cf - size\ P,\ snd\ cf)$

Next, some lemmas about *execl* and *execl-all* are proven. In more detail, given a program P and a list of configurations *cfs* such that $P \models cfs$:

- Lemma *execl-next* states that for any configuration in *cfs* but the last one, the subsequent configuration must result from the execution of the referenced instruction of P in that configuration.
  Thus, *execl-next* permits to reproduce the execution of a single instruction.
- Lemma *execl-last* states that a configuration in *cfs* whose program counter falls outside P must be the last one in *cfs*.
  Thus, *execl-last* permits to infer the completion of program execution.
- Lemma *execl-drop* states that $P \models drop\ n\ cfs$ for any natural number n, and will be used to prove compiler correctness for loops by induction over the length of the list of configurations *cfs*.

Furthermore, some other lemmas enabling to prove compiler correctness automatically for constructors *N*, *V* (arithmetic expressions), *Bc* (boolean expressions) and *SKIP* (commands) are also proven.

lemma *iexec-offset-aux*:
$(i :: int) + 1 - j = i - j + 1 \wedge i + j - k + 1 = i - k + j + 1$

⟨*proof*⟩

lemma *iexec-offset* [*intro*]:
(*ins*, *pc*, *s*, *stk*) ↦ (*pc′*, *s′*, *stk′*) ⟹
(*ins*, *pc* − *i*, *s*, *stk*) ↦ (*pc′* − *i*, *s′*, *stk′*)
⟨*proof*⟩

lemma *execl-next*:
⟦*P* |= *cfs*; *k* < *length cfs*; *k* ≠ *length cfs* − *1*⟧ ⟹
(*P* !! *fst* (*cfs* ! *k*), *cfs* ! *k*) ↦ *cfs* ! *Suc k*
⟨*proof*⟩

lemma *execl-last*:
⟦*P* |= *cfs*; *k* < *length cfs*; *fst* (*cfs* ! *k*) ∉ {*0*..<*size P*}⟧ ⟹
length cfs − *1* = *k*
⟨*proof*⟩

lemma *execl-drop*:
P |= *cfs* ⟹ *P* |= *drop n cfs*
⟨*proof*⟩

lemma *execl-all-N* [*simplified*, *intro*]:
[*LOADI i*] |= *cfs*□ ⟹ *apred* (*N i*) (*cfs* ! *0*) (*cfs* ! (*length cfs* − *1*))
⟨*proof*⟩

lemma *execl-all-V* [*simplified*, *intro*]:
[*LOAD x*] |= *cfs*□ ⟹ *apred* (*V x*) (*cfs* ! *0*) (*cfs* ! (*length cfs* − *1*))
⟨*proof*⟩

lemma *execl-all-Bc* [*simplified*, *intro*]:
⟦*if v* = *f then* [*JMP i*] *else* [] |= *cfs*□; *0* ≤ *i*⟧ ⟹
bpred (*Bc v*, *f*, *i*) (*cfs* ! *0*) (*cfs* ! (*length cfs* − *1*))
⟨*proof*⟩

lemma *execl-all-SKIP* [*simplified*, *intro*]:
[] |= *cfs*□ ⟹ *cpred SKIP* (*cfs* ! *0*) (*cfs* ! (*length cfs* − *1*))
⟨*proof*⟩

Next, lemma *execl-all-sub* is proven. It states that, if *P* @ *P′ x* @ *P″* |= *cfs*□, configuration *cf* within *cfs* refers to the start of program *P′ x*, and the initial and the final configuration in every complete execution of *P′ x* satisfy predicate *Q x*, then there exists a configuration *cf′* in *cfs* such that *cf* and *cf′* satisfy *Q x*.
Thus, this lemma permits to reproduce the execution of a subprogram, particularly:

- a compiled arithmetic expression *a*, where *Q* = *apred* and *x* = *a*,

- a compiled boolean expression b, where $Q = bpred$ and $x = (b, f, i)$ (given a flag f and a jump offset i), and
- a compiled command c, where $Q = cpred$ and $x = c$.

Furthermore, lemma *execl-all-sub2* is derived from *execl-all-sub* to enable a shorter symbolical execution of two consecutive subprograms.

lemma *execl-sub-aux*:
$[\![\bigwedge m\ n.\ \forall k \in \{m..<n\}.\ Q\ P'\ (((pc,\ s,\ stk)\ \#\ cfs)\ !\ k) \Longrightarrow P' \models$
$map\ (off\ P)\ (case\ m\ of\ 0 \Rightarrow (pc,\ s,\ stk)\ \#\ take\ n\ cfs \mid Suc\ m \Rightarrow F\ cfs\ m\ n);$
$\forall k \in \{m..<n+m+length\ cfs'\}.\ Q\ P'\ ((cfs' @ (pc,\ s,\ stk)\ \#\ cfs)\ !\ (k-m))]\!] \Longrightarrow$
$P' \models (pc - size\ P,\ s,\ stk)\ \#\ map\ (off\ P)\ (take\ n\ cfs)$
(**is** $[\![\bigwedge$- -. $\forall k \in$ -. $Q\ P'\ (?F\ k) \Longrightarrow$ -; $\forall k \in ?A.\ Q\ P'\ (?G\ k)]\!] \Longrightarrow$ -)
⟨*proof*⟩

lemma *execl-sub*:
$[\![P @ P' @ P'' \models cfs;\ \forall k \in \{m..<n\}.$
$size\ P \le fst\ (cfs\ !\ k) \wedge fst\ (cfs\ !\ k) - size\ P < size\ P']\!] \Longrightarrow$
$P' \models map\ (off\ P)\ (drop\ m\ (take\ (Suc\ n)\ cfs))$
(**is** $[\![$-; $\forall k \in$ -. $?P\ P'\ cfs\ k]\!] \Longrightarrow P' \models map$ - ($?F\ cfs\ m\ (Suc\ n)$))
⟨*proof*⟩

lemma *execl-all-sub* [*rule-format*]:
assumes
A: $P @ P' x @ P'' \models cfs\Box$ **and**
B: $k < length\ cfs$ **and**
C: $fst\ (cfs\ !\ k) = size\ P$ **and**
D: $\forall cfs.\ P'\ x \models cfs\Box \longrightarrow Q\ x\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
shows $\exists k' < length\ cfs.\ Q\ x\ (off\ P\ (cfs\ !\ k))\ (off\ P\ (cfs\ !\ k'))$
⟨*proof*⟩

lemma *execl-all-sub2*:
assumes
A: $P\ x @ P'\ x' @ P'' \models cfs\Box$
(**is** $?P \models$ -$\Box$) **and**
B: $\bigwedge cfs.\ P\ x \models cfs\Box \Longrightarrow (\lambda(pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc' = pc + size\ (P\ x) + I\ s \wedge Q\ s\ s' \wedge stk' = F\ s\ stk)$
$(cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
(**is** $\bigwedge cfs.$ - $\Longrightarrow ?Q\ x\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$) **and**
C: $\bigwedge cfs.\ P'\ x' \models cfs\Box \Longrightarrow (\lambda(pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc' = pc + size\ (P'\ x') + I'\ s \wedge Q'\ s\ s' \wedge stk' = F'\ s\ stk)$
$(cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
(**is** $\bigwedge cfs.$ - $\Longrightarrow ?Q'\ x'\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$) **and**
D: $I\ (fst\ (snd\ (cfs\ !\ 0))) = 0$
shows $\exists k < length\ cfs.\ \exists t.\ (\lambda(pc,\ s,\ stk)\ (pc',\ s',\ stk').$
$pc = 0 \wedge pc' = size\ (P\ x) + size\ (P'\ x') + I'\ t \wedge Q\ s\ t \wedge Q'\ t\ s' \wedge$
$stk' = F'\ t\ (F\ s\ stk))\ (cfs\ !\ 0)\ (cfs\ !\ k)$
⟨*proof*⟩

2.2 Main theorem

It is time to prove compiler correctness. First, lemmas *acomp-acomp*, *bcomp-bcomp* are derived from *execl-all-sub2* to reproduce the execution of two consecutive compiled arithmetic expressions (possibly generated by both *acomp* and *bcomp*) and boolean expressions (possibly generated by *bcomp*), respectively. Subsequently, the correctness of *acomp* and *bcomp* is proven in lemmas *acomp-correct*, *bcomp-correct*.

lemma *acomp-acomp*:
$[\![acomp\ a_1\ @\ acomp\ a_2\ @\ P \models cfs\Box;$
$\quad \bigwedge cfs.\ acomp\ a_1 \models cfs\Box \Longrightarrow apred\ a_1\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1));$
$\quad \bigwedge cfs.\ acomp\ a_2 \models cfs\Box \Longrightarrow apred\ a_2\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))]\!] \Longrightarrow$
$case\ cfs\ !\ 0\ of\ (pc,\ s,\ stk) \Rightarrow pc = 0 \wedge (\exists k < length\ cfs.\ cfs\ !\ k =$
$\quad (size\ (acomp\ a_1\ @\ acomp\ a_2),\ s,\ aval\ a_2\ s\ \#\ aval\ a_1\ s\ \#\ stk))$
⟨*proof*⟩

lemma *bcomp-bcomp*:
$[\![bcomp\ (b_1,\ f_1,\ i_1)\ @\ bcomp\ (b_2,\ f_2,\ i_2) \models cfs\Box;$
$\quad \bigwedge cfs.\ bcomp\ (b_1,\ f_1,\ i_1) \models cfs\Box \Longrightarrow$
$\quad\quad bpred\ (b_1,\ f_1,\ i_1)\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1));$
$\quad \bigwedge cfs.\ bcomp\ (b_2,\ f_2,\ i_2) \models cfs\Box \Longrightarrow$
$\quad\quad bpred\ (b_2,\ f_2,\ i_2)\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))]\!] \Longrightarrow$
$case\ cfs\ !\ 0\ of\ (pc,\ s,\ stk) \Rightarrow pc = 0 \wedge (bval\ b_1\ s \neq f_1 \longrightarrow$
$\quad (\exists k < length\ cfs.\ cfs\ !\ k = (size\ (bcomp\ (b_1,\ f_1,\ i_1)\ @\ bcomp\ (b_2,\ f_2,\ i_2)) +$
$\quad\quad (if\ bval\ b_2\ s = f_2\ then\ i_2\ else\ 0),\ s,\ stk)))$
⟨*proof*⟩

lemma *acomp-correct* [*simplified*, *intro*]:
$acomp\ a \models cfs\Box \Longrightarrow apred\ a\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
⟨*proof*⟩

lemma *bcomp-correct* [*simplified*, *intro*]:
$[\![bcomp\ x \models cfs\Box;\ 0 \leq snd\ (snd\ x)]\!] \Longrightarrow bpred\ x\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
⟨*proof*⟩

Next, lemmas *bcomp-ccomp*, *ccomp-ccomp* are derived to reproduce the execution of a compiled boolean expression followed by a compiled command and of two consecutive compiled commands, respectively (possibly generated by *ccomp*). Then, compiler correctness for loops and for all commands is proven in lemmas *while-correct* and *ccomp-correct*, respectively by induction over the length of the list of configurations and by structural induction over commands.

lemma *bcomp-ccomp*:
$[\![bcomp\ (b,\ f,\ i)\ @\ ccomp\ c\ @\ P \models cfs\Box;\ 0 \leq i;$

$\bigwedge cfs.\ ccomp\ c \models cfs\Box \Longrightarrow cpred\ c\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))\rrbracket \Longrightarrow$
$case\ cfs\ !\ 0\ of\ (pc,\ s,\ stk) \Rightarrow pc = 0 \wedge (bval\ b\ s \neq f \longrightarrow$
$(\exists k < length\ cfs.\ case\ cfs\ !\ k\ of\ (pc',\ s',\ stk') \Rightarrow$
$pc' = size\ (bcomp\ (b,\ f,\ i)\ @\ ccomp\ c) \wedge (c,\ s) \Rightarrow s' \wedge stk' = stk))$
⟨*proof*⟩

lemma *ccomp-ccomp*:
$\llbracket ccomp\ c_1\ @\ ccomp\ c_2 \models cfs\Box;$
$\bigwedge cfs.\ ccomp\ c_1 \models cfs\Box \Longrightarrow cpred\ c_1\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1));$
$\bigwedge cfs.\ ccomp\ c_2 \models cfs\Box \Longrightarrow cpred\ c_2\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))\rrbracket \Longrightarrow$
$case\ cfs\ !\ 0\ of\ (pc,\ s,\ stk) \Rightarrow pc = 0 \wedge (\exists k < length\ cfs.\ \exists t.$
$case\ cfs\ !\ k\ of\ (pc',\ s',\ stk') \Rightarrow pc' = size\ (ccomp\ c_1\ @\ ccomp\ c_2) \wedge$
$(c_1,\ s) \Rightarrow t \wedge (c_2,\ t) \Rightarrow s' \wedge stk' = stk)$
⟨*proof*⟩

lemma *while-correct* [*simplified*, *intro*]:
$\llbracket bcomp\ (b,\ False,\ size\ (ccomp\ c) + 1)\ @\ ccomp\ c\ @$
$[JMP\ (-\ (size\ (bcomp\ (b,\ False,\ size\ (ccomp\ c) + 1)\ @\ ccomp\ c) + 1))] \models$
$cfs\Box;$
$\bigwedge cfs.\ ccomp\ c \models cfs\Box \Longrightarrow cpred\ c\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))\rrbracket \Longrightarrow$
$cpred\ (WHILE\ b\ DO\ c)\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - Suc\ 0))$
(**is** $\llbracket ?cb\ @\ ?cc\ @\ [JMP\ (-\ ?n)] \models \text{-}\Box;\ \bigwedge \text{-}.\ \text{-} \Longrightarrow \text{-}\rrbracket \Longrightarrow ?Q\ cfs$)
⟨*proof*⟩

lemma *ccomp-correct*:
$ccomp\ c \models cfs\Box \Longrightarrow cpred\ c\ (cfs\ !\ 0)\ (cfs\ !\ (length\ cfs - 1))$
⟨*proof*⟩

Finally, the main compiler correctness theorem, expressed using predicate *exec*, is proven. First, $P \vdash cf \rightarrow* cf'$ is shown to imply the existence of a nonempty list of configurations *cfs* such that $P \models cfs$, whose initial and final configurations match *cf* and *cf'*, respectively. Then, the main theorem is derived as a straightforward consequence of this lemma and of the previous lemma *ccomp-correct*.

lemma *exec-execl* [*dest*!]:
$P \vdash cf \rightarrow* cf' \Longrightarrow \exists cfs.\ P \models cfs \wedge cfs \neq [] \wedge hd\ cfs = cf \wedge last\ cfs = cf'$
⟨*proof*⟩

theorem *ccomp-exec*:
$ccomp\ c \vdash (0,\ s,\ stk) \rightarrow* (size\ (ccomp\ c),\ s',\ stk') \Longrightarrow (c,\ s) \Rightarrow s' \wedge stk' = stk$
⟨*proof*⟩

end

References

[1] G. Klein. Theory HOL-IMP.Compiler2 (included in the Isabelle2021 distribution). https://isabelle.in.tum.de/website-Isabelle2021/dist/library/HOL/HOL-IMP/Compiler2.html.

[2] A. Krauss. *Defining Recursive Functions in Isabelle/HOL*. https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/functions.pdf.

[3] T. Nipkow. *A Tutorial Introduction to Structured Isar Proofs*. https://isabelle.in.tum.de/website-Isabelle2011/dist/Isabelle2011/doc/isar-overview.pdf.

[4] T. Nipkow. *Programming and Proving in Isabelle/HOL*, Feb. 2021. https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/prog-prove.pdf.

[5] T. Nipkow and G. Klein. *Concrete Semantics with Isabelle/HOL*. Springer-Verlag, Mar. 2021. (Current version: http://www.concrete-semantics.org/concrete-semantics.pdf).

[6] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL – A Proof Assistant for Higher-Order Logic*, Feb. 2021. https://isabelle.in.tum.de/website-Isabelle2021/dist/Isabelle2021/doc/tutorial.pdf.